
REjercicio TC32-39p

En varias ramas de la F́ısica, especialmente en Óptica, aparece la necesidad de calcular
repetidamente la función conocida como Seno Integral, Si(x), definida como

Si(x) =

∫
x

0

sen(t)

t
dt =

∞∑
n=0

(−1)n · x2n+1

(2n + 1) · (2n + 1)!
, x > 0.

Hay que escribir una rutina, en lenguaje ensamblador del 80x86/x87, que calcule la función
Seno Integral según el desarrollo en serie indicado. A la rutina se le deben pasar por la pila los
siguientes parámetros:

argumento de la función, x,

error máximo admisible en el cálculo de la función, err,

número máximo de iteraciones permitidas, N ,

de la forma
SenoIntegral(x, err, N)

siguiendo el convenio del lenguaje C de paso de parámetros, primero el de más a la derecha y se
termina con el de más a la izquierda. Por ejemplo, se llamaŕıa a la rutina aśı:

push N

push err

push x

call SenoIntegral

Si se logra calcular la función en menos de N iteraciones con error menor que err el resultado se
devuelve en el registro ST(0) de la FPU y el número de iteraciones realizadas en AX, poniendo
CF=0. En caso de agotar el número de iteraciones N sin lograr un error menor de err se devuelve
la aproximación obtenida en ST(0) y el indicador de carry, CF=1, para señalar la no convergencia.

Solución

La serie

Si(x) =
∞∑

n=0

(−1)n · x2n+1

(2n + 1) · (2n + 1)!
= x −

x3

3 · 3!
+

x5

5 · 5!
−

x7

7 · 7!
+ . . .

se calculará iterativamente, sumando/restando un término en cada iteracción. Un primer esque-
ma de cálculo seŕıa el que se plantea en el siguiente diagrama de flujo:

k = 0
S = 0.0

ak = (−1)k · x2k+1/(2k + 1)/(2k + 1)!
S = S + ak

|ak| < err

k = k + 1 CF = 0

k > N

CF = 1

ya que, al ser una serie alterna, el error es menor que el nuevo término que se añade:

err < ǫk = |Sk − Sk−1| = |ak|

Los términos de la serie pueden calcularse recurrentemente:

ak+1

ak

=

(−1)k+1
·x

2k+3

(2k+3)·(2k+3)!

(−1)k
·x2k+1

(2k+1)·(2k+1)!

= −
(2k + 1) · x2

(2k + 3)(2k + 2)(2k + 3)

O sea:

ak+1 = − (2k+1)·x2

(2k+2)·(2k+3)2 · ak k = 0, 1, . . .

Los coeficientes son:
(2k + 1)

(2k + 2)(2k + 3)2

1
2·32 , 3

4·52 , 5
6·72 , . . . que pueden irse calculando por su cuenta aśı: n0 = 1, nk+1 = nk + 2,

ck =
nk

(nk + 1) · (nk + 2)2
.

Poniendo todo esto junto:

k = 0
a = x
n = 1
S = a

k + +
c = n/(n + 1)/(n + 2)2

a = −c · x2 · a
S = S + a

|a| < err

n = n + 2 CF = 0

k > N

CF = 1

FIN

FIN

Se precisan dos contadores enteros, k y n. El primero no participa de los cálculos y se puede
mantener en un registro. El segundo, n, hay que pasárselo a la FPU y conviene mantenerlo en
memoria, mejor en memoria local de la rutina.

Los valores en coma flotante que hay que mantener de un ciclo al siguiente son: el parámetro
de entrada x, el término a y la suma parcial S. Los tres valores pueden mantenerse en los
registros de la FPU. El coeficiente c se calcula en cada iteración.

Conviene que el orden de los valores en los registros de la FPU sea

a st(0)
x st(1)
S st(2)
. . .

al comenzar el bucle de la iteración, evolucionando aśı el contenido de los registros de la FPU:

Rst(0) st(1) st(2) st(3) -

n a x S

n/(n + 1) a x S

a · n/(n + 1) x S

x a · n/(n + 1) x S

x/(n + 2) a · n/(n + 1) x S

x/(n + 2) (x/(n + 2)) · (a · n/(n + 1)) x S

a = (x/(n + 2)) · (x/(n + 2)) · (a · n/(n + 1)) x S

a = −a x S

a x S = S + a

a a x S

|a| a x S

y se compara |a| con err, quitando a |a| y quedando la pila como al principio.
El código de la rutina puede ser el siguiente:

SenoIntegral proc near

push bp

mov bp,sp

sub sp,2 ; hace sitio para n en la pila

; ***

fld DWORD PTR [bp+4] ; x |

xor ax,ax ; k = 0

mov WORD PTR [bp-2],1 ; n = 1

fld st(0) ; x | S = x |

fld st(0) ; a = x | x | S |

@bucle:

inc ax

fild WORD PTR [bp-2] ; n | a | x | S |

inc WORD PTR [bp-2] ; n+1

fidiv WORD PTR [bp-2] ; n/(n+1) | a | x | S |

inc WORD PTR [bp-2] ; n+2

fmul ; a·n/(n+1) | x | S |

fld st(1) ; x | a·n/(n+1) | x | S |

fidiv WORD PTR [bp-2] ; x/(n+2) | a·n/(n+1) | x | S |

fmul st(1),st(0)

; x/(n+2) | (x/(n+2))·(a·n/(n+1)) | x | S |

fmul ; a’ | x | S |

fchs ; a | x | S |

fadd st(2),st(0) ; a | x | S=S+a |

fld st(0) ; a | a | x | S |

fabs ; abs(a) | a | x | S |

fcomp DWORD PTR [bp+8] ; ¿abs(a) < err?

push ax

fstsw ax

sahf

pop ax

jb @converge

cmp ax,WORD PTR [bp+12] ; ¿k > N?

Rja @no_converge

jmp @bucle

@converge:

clc ; CF = 0

jmp @final

@no_converge:

stc ; CF = 1

@final:

fstp st(0)

fstp st(0)

mov sp,bp

pop bp

ret

SenoIntegral endp

La pila en el punto *** del programa es:

N

err

x

direc. vuelta

BP antiguo

n

libre

SP > −2

< BP

+2

+4

+8

+12

